

DK Wildfinity

Een trainer voor wild barrels. Net zoals Springfinity start je het barrel board en daar blijf je dan tot in de oneindigheid in hangen. Gewoon op level L=01. Maar daarna ervoor zorgen dat de bonus timer niet afloopt en de difficulty niet aangepast wordt.

Dan weergeven en instelbaarmaken internal difficulty met P1:

- Difficulty 1 – Wild type 1
- Difficulty 2 – Wild type 1
- Difficulty 3 – Wild type 2
- Difficulty 4 – Wild type 2
- Difficulty 5 – Wild type 3

Dan weergeven en instelbaar maken wild frequency met P2:

- 1x = normal frequency
- 2x = two times higher chance
- 4x = four times higher chance
- 8x = eight times higher chance
- Wilds only – met daarbij dan de deploysnelheid iets verlagen

Mogelijkheid om hamers te herstellen met lang indrukken P1:

Mogelijkheid om de fireballs te verwijderen met lang indrukken P2:

Eerst beginnen met de standaard DK rom.

Als dood dan geluid afspelen en doorgaan

Komt eigenlijk neer op de invincibility. Maar dan niet met een NOP van de jump in de main, maar wel springen, maar alleen testen en wanneer dood dan alleen geluid afspelen en returnen.

In de main wordt dit hier geregeld:

```
19B3 CD0828 CALL #2808 ; check for collisions with hostile sprites [set to NOPS to make mario invincible to enemy sprites]
```

Dus het stukje code voor de afhandeling begint op #2808.

```
; called from main routine at #19B3
; checks for collisions with hostiles sprites

2808 FD210062 LD IY,#6200 ; load IY with start of mario sprite
280C 3A0562 LD A,(#6205) ; load A with mario's Y position
280F 4F LD C,A ; copy to C
2810 210704 LD HL,#0407 ; H := 4, L := 7
2813 CD6F28 CALL #286F ; checks for collisions based on the screen. A := 1 if collision, otherwise zero
2816 A7 AND A ; was there a collision ?
2817 C8 RET Z ; no, return

; mario collided with hostile sprite

2818 3D DEC A ; else A := 0
2819 320062 LD (#6200),A ; store into mario life indicator, mario is dead
281C C9 RET ; return
```

Dus bij #2819 springen naar een stukje code die het dead muziekje start en daarna de return doet, zonder de life indicator te zetten.

Dat zou dit stukje code kunnen zijn:

```

12A6 3E03 LD A,#03 ; load A with duration of sound
12A8 328860 LD (#6088),A ; play death sound
12AB C9 RET ; return

```

Dus dan aanpassen:

```

2819 C3A612 JP #12A6 ; play death sound, and return

```

Als bij Pauline dan geluid spelen, maar niet einde level

```

1E6D 3E00 LD A,#00 ; load A with #00. sprite for facing left
1E6F DA741E JP C,#1E74 ; if on left side, skip next step

1E72 3E80 LD A,#80 ; else load A with sprite facing right
1E74 324D69 LD (#694D),A ; set mario sprite
1E77 C3851E JP #1E85 ; jump ahead

; check for end of level on girders and elevators

1E7A FE31 CP #31 ; are we on top level (rescued girl?)
1E7C D0 RET NC ; no, return

1E7D C36D1E JP #1E6D ; level has been fished. jump to end of level routine.

; arrive here when on rivets

1E80 3A9062 LD A,(#6290) ; load A with number of rivets left
1E83 A7 AND A ; all done with rivets ?
1E84 C0 RET NZ ; no, return

1E85 3E16 LD A,#16 ; else A := #16
1E87 320A60 LD (GameMode2),A ; store into game mode2
1E8A E1 POP HL ; pop stack to get higher address
1E8B C9 RET ; return to a higher level [returns to #00D2]

```

Gaat om barrels. Dus volgorde van de jumps: #1E7A -> #1E6D -> #1E85.

Dus bij #1E85 niet de gamemode2 aanpassen, maar alleen een geluid spelen en dan een RET.

Dat is eigenlijk het stukje vanaf #1729.

Aanpassen:

```

1E7D C32917 JP #1729 ; only play end of level music, and return (=continue game)

```

```

1729 218A60 LD HL,#608A ; load sound address
172C 3607 LD (HL),#07 ; play sound for end of level
172E 23 INC HL ; HL now has sound duration
172F 3603 LD (HL),#03 ; set duration to 3
1731 C9 RET ; return

```

Ja dat gaat goed.

Alleen op de girder waar Pauline staat wordt ook nog de aanpassing van jumpman's Y-positie t.a.v. schuine girders gedaan. Dat hoeft niet. **Later nog aanpassen omdat we anders wellicht**

geheugenplaatsen gebruiken die in Springfinito ook gebruikt worden en we niet één-op-één kunnen kopiëren. Waarschijnlijk de check op 234D vervangen door een dubbele check: als kleiner dan #40 dan return en als #4D dan origineel.

De bonus timer niet terug laten lopen

Dat gebeurt hier:

```
2fe5 21B162 LD HL,#62B1 ; load HL with bonus timer
2fe8 35 DEC (HL) ; Decrement. is the bonus timer zero?
2fe9 c0 RET NZ ; no, return

2fea 3E01 LD A,#01 ; else time has run out. A := 1
2fec 328663 LD (#6386),A ; set time has run out indicator
2fef c9 RET ; return
```

Aanpassen:

```
2FE8 C9 RET ; return without decrementing bonus timer
```

Raar. Je gaat niet dood door de bonus timer. Dus dat is goed, maar de timer loopt nog wel af.

Er is zoiets als een 'on screen' timer:

```
; jump here from #0632

06A8 d601 SUB #01 ; subtract 1 from bonus timer
06Aa 2005 JR NZ,#06b1 ; If not zero, skip next 2 steps

; timer at zero

06Ac 21B863 LD HL,#63B8 ; load HL with mario dead flag
06Af 3601 LD (HL),#01 ; store 1 - mario will die soon on next timer click

06b1 27 DAA ; Decimal adjust
06b2 328C63 LD (#638C),A ; store A into timer
06b5 c36A06 JP #066A ; jump back
```

Aanpassen:

```
06A8 0000 NOP,NOP
06AA C36A06 JP #066A ; jump back
```

Dat gaat niet goed. On screen blijft de bonus timer gewoon stil staan. Maar je gaat wel dood na een bepaalde tijd. Bij barrels wordt de bonus timer bepaald door de barrels. Dus dat is waarschijnlijk een ander stuk code dan voor elevators. Later nog aanpassen omdat we anders wellicht geheugenplaatsen gebruiken die in Springfinito ook gebruikt worden en we niet één-op-één kunnen kopiëren.

Titel aanpassen in titelscherm en High Score

Nu de High Score tekst en het titelscherm aanpassen.

CREDIT opschuiven:

#36C2: 3F 9F 75 13 22 15 14 19 24 3F

Aanpassen verwijzing:

3655: C2 36 5 ; #36C2 "CREDIT"

en

HIGH SCORE aanpassen naar WILDFINITY

#36B2: 60 76 27 19 1C 14 16 19 1E 19 24 29 3F

Het titelscherm aanpassen

De KONG letters 1 positie naar beneden verschuiven:

3D59: 05 30 77 05 10 77 02 F1 76 02 D0 76 02 D3 76 ; K
3D68: 05 90 76 05 70 76 01 50 76 01 54 76 05 30 76 ; O
3D77: 05 F0 75 02 D1 75 02 B2 75 05 90 75 ; N
3D83: 03 51 75 05 30 75 01 10 75 01 14 75 ; G (part 1)
3D8F: 01 F0 74 01 F2 74 01 F4 74 02 D2 74 ; G (part 2)
3D9B: 00 ; end code

En het TM symbool weghalen.

```
081C CD243F CALL #3F24 ; draw TM logo onscreen [patch? orig japanese had 3 NOPs here]
```

Hiermee wordt het stukje extra code dat het TM-symbool tekent ook vrijgemaakt:

#3F24 t/m #3F2F.

Dus op #081C weer 3 NOP's maken.

Dan kan het stuk van #3F00 t/m #3F23 ook vrijgemaakt worden en gebruikt worden voor de teksten in het titelscherm.

```
3687: 00 3F 1E ; #3F00 "(C) 1981"  
3689: 09 3F 1F ; #3F09 "NINTENDO OF AMERICA"
```

In #3F00 de copyright notice maken:

© 1981-2023 NINTENDO

5C 77 49 4A 10 01 09 08 01 2C 02 00 02 03 10 1E 19 1E 24 15 1E 14 1F 3F

En de plek van de checksum ook aanpassen: INTEND: start op 3F10:

2441 21103F LD HL,#3F10 ; load HL with ROM area that has NINTENDO written

En dan in #3F18 de titel maken:
WILDFINITY

8E 76 27 19 1C 14 16 19 1E 19 24 29 3F

En de verwijzing ernaar toe aanpassen:

3689: 18 3F

Nu nog de Donkey Kong een regel naar beneden verplaatsen.

Wordt hier gedaan:

```
0820 210869 LD HL,#6908 ; load HL with start of kong sprite X pos
0823 0E44 LD C,#44 ; load C with offset to add X
0825 FF RST #38 ; draw kong in new position
0826 210869 LD HL,#6908 ; load HL with start of kong sprite Y pos
0829 0E78 LD C,#78 ; load C with offset to add Y
082B FF RST #38 ; draw kong
082C C9 RET ; return
```

Kan door aanpassen:

0829: 0E80 LD C,#80 ; load C with offset to add Y

Niet te testen omdat de attract mode nooit eindigt omdat jumpman niet dood gaat...

Meteen beginnen met het titelscherm

Aanpassen:

01ED C3983F JP #3F98 ; jump to additional code – jump to label AAAA
01F0 00 NOP

en

```
3F98 3E06 LD A,#06 ; A := 6 – label AAAA
3F9A 320A60 LD (GameMode2),A ; store into game mode 2
3F9D C3F101 JP #01F1 ; jump back
```

Versienummer toevoegen

Liefste een versie toevoegen aan het titel scherm. Eerste versie opnemen is v1.00.

De 1 is nog unused:

364B:	8B 36	0	; #368B "GAME OVER"
364D:	01 00	1	; unused ?
364F:	98 36	2	; #3698 "PLAYER <I>"
3651:	A5 36	3	; #36A5 "PLAYER <II>"
3653:	B2 36	4	; #36B2 "HIGH SCORE"
3655:	BF 36	5	; #36BF "CREDIT"
3657:	06 00	6	; unused ?
3659:	CC 36	7	; #36CC "HOW HIGH CAN YOU GET?"

Deze laten verwijzen naar het stukje met spaties in de Name:

```

3710: 24 24 1F 1E 3F 27 76 20 25 23 18 3F 06 77 1E 11 TTON...PUSH...NA
3720: 1D 15 10 22 15 17 19 23 24 22 11 24 19 1F 1E 3F ME.REGISTRATION.
3730: 88 76 1E 11 1D 15 2E 3F E9 75 2D 2D 2D 10 10 10 ..NAME:....
3740: 10 10 10 10 10 10 3F 0B 77 11 10 12 10 13 10 14 .....A.B.C.D
3750: 10 15 10 16 10 17 10 18 10 19 10 1A 3F 0D 77 1B .E.F.G.H.I.J...K
3760: 10 1C 10 1D 10 1E 10 1F 10 20 10 21 10 22 10 23 .L.M.N.O.P.Q.R.S
3770: 10 24 3F 0F 77 25 10 26 10 27 10 28 10 29 10 2A .T...U.V.W.X.Y.Z

```

Dus daartoe #373D veranderen in 3F.

En dan de versie opnemen in stuk daarna:

#373E: 3D 76 26 01 2B 00 00 10 3F (V1.00)

en

#364D: 3E 37

We hadden de #081C drie NOP's gemaakt (niet meer TM logo tekenen). Daar kan dan mooi een aanroep naar een stuk additionele code:

081C: C3BA3F JP #3FBA ; jump to additional code to display version

3FBA: 110103 LD DE,#0301 ; load task data for text "V1.00"

3FBD: CD9F30 CALL #309F ; insert task to draw text

3FC0: C31F08 JP #081F ; jump back

Ja dat is goed.

De animatie waar Kong de trappen met Pauline beklimt overslaan

Dit stuk begint bij 0A76. Hele stuk niet doen, alleen de gamemode2 verhogen:

0A76 210A60 LD HL,GameMode2

0A79 34 INC (HL) ; increase game mode2 (to 8?)

0A7A C9 RET

Ja dat gaat ook goed.

Aanpassen tekst How High Can You Get?

De tekst HOW HIGH CAN YOU GET ? veranderen in LET'S PRACTICE !
Korter dus aanpassen op originele plek #36CC:

FE 76 1C 15 24 3A 23 10 20 22 11 13 24 19 13 15 10 38 3F

Ja dat is ook goed.

Nu versie v0.01 opgeslagen.

Attract mode uitzetten

Wordt gestuurd door gamemode2. Die wordt in begin al op 6 gezet om te beginnen met het titelscherm met het Donkey Kong logo dat knippert.

```
0746 7E LD A,(HL) ; else load A with game mode2
0747 EF RST #28 ; jump based on A

0748 79 07 0 ; #0779 ; clear screen, set color palettes, draw attract mode text and high score table,
; [continued] increase game mode2, clear sprites, ; draw "IUP" on screen , draws number of coins needed for play
074A 63 07 1 ; #0763 ;
074C 3C 12 2 ; #123C ; set initial mario sprite position and draw remaining lives and level
074E 77 19 3 ; #1977 ; set artificial input for demo play [change to #197A to enable playing in demo part 1/2]
0750 7C 12 4 ; #127C ; handle mario dying animations
0752 C3 07 5 ; #07C3 ; clears the screen and sprites and increase game mode2
0754 CB 07 6 ; #07CB ; handle intro splash screen ?
0756 4B 08 7 ; #084B ; counts down a timer then resets game mode2 to 0

0758 00 00 00 00 ; unused
```

In gamemode2 = 7 wordt alleen een timer gezet en gewacht en wordt daarna de gamemode2 terug op 0 gezet. Als we die daar terug zetten op 6 dan blijft het logo zichtbaar en gaat het opnieuw knipperen. Dus dan geen high score scherm en geen attract mode meer. Is prima.

```
; arrive from #0747 when GameMode2 == 7
```

```
084B E7 RST #20 ; update timer and continue here only when complete, else RET

084C 210A60 LD HL,GameMode2 ; load HL with game mode2
084F 3600 LD (HL),#00 ; set to 0
0851 C9 RET ; return
```

Aanpassen:

```
084F 3606 LD (HL),#06 ; set to 6
```

Dat gaat goed.

Wanneer dood dan geen aantal jumpman verlagen.

Dat zit hier:

```

; player 1 died
; clear sounds, decrease life, check for and handle game over

12F2 CD1C01 CALL #011C ; clear all sounds
12F5 AF XOR A ; A := 0
12F6 322C62 LD (#622C),A ; store into game start flag
12F9 212862 LD HL,#6228 ; load HL with address for number of lives remaining
12FC 35 DEC (HL) ; one less life
12FD 7E LD A,(HL) ; load A with number of lives left
12FE 114060 LD DE,P1NumLives ; set destination address
1301 010800 LD BC,#0008 ; set counter
1304 EDB0 LDIR ; copy (#6228) to (#6230) into (P1NumLives) to (P2NumLives).
1306 A7 AND A ; number of lives == 0 ?
1307 C23413 JP NZ,#1334 ; no, skip ahead

```

Aanpassen:

```
12FC 00 NOP
```

Ja dat gaat goed.

Aangeven oneindig aantal levens

Aantal levens wordt hier getekend:

```

06C7 3A2862 LD A,(#6228) ; load A with number of lives remaining
06CA 91 SUB C ; subtract the task parameter. zero lives to draw?
06CB CAD706 JP Z,#06D7 ; yes, skip next 5 steps

06CE 47 LD B,A ; For B = 1 to A
06CF 218377 LD HL,#7783 ; load HL with screen location to draw remaining lives

06D2 36FF LD (HL),#FF ; draw the extra mario
06D4 19 ADD HL,DE ; add offset for next column
06D5 10FB DJNZ #06D2 ; next B

```

Dit aanpassen naar het volgende:

```

06C7 218377 LD HL,#7783 ; load HL with screen location for remaining lives
06CA 36FF LD (HL),#FF ; draw the jumpman
;
06CC 19 ADD HL,DE ; add offset for next column
06CD 3634 LD (HL),#34 ; draw the equal sign
;
06CF 19 ADD HL,DE ; add offset for next column
06D0 3631 LD (HL),#BF ; draw the infinity sign
06D2 0000000000 5x NOP's

```


Ja dat werkt goed.

Nu versie v0.02 opgeslagen.

Nu eventjes de twee eerdere problemen oplossen.

Bonus Timer niet verlagen in barrels

Bonus timer in barrels wordt op een andere plek verlaagd dan in elevators. Dus de eerder gedane aanpassing is niet zo zinvol en heeft geen effect.

Voor barrels wordt dat hier gedaan:

```
2CDA 21B162 LD HL,#62B1 ; load bonus counter into HL
2CDD 35 DEC (HL) ; decrement bonus counter. Is it zero?
2CDE c2E62C JP NZ,#2Ce6 ; no, skip next 2 steps

2CE1 3E01 LD A,#01 ; A := 1
2CE3 328663 LD (#6386),A ; store into bonus timer out indicator

2CE6 7E LD A,(HL) ; load A with bonus counter
```

Dit op te lossen door het stukje #2CE1 t/m 2CE5 altijd over te slaan. Door de jump onconditioneel uit te voeren.

Aanpassen:

```
2CDE C3E62C JP #2CE6 ; always skip next 2 steps
```

Gaat niet goed. Maar wel grappig. Donkey Kong rolt de laatste barrels en de stapel raakt op. En daarna blijft hij wel bewegen, maar worden er geen barrels meer gegooid.

De bonus counter moet ook niet meer verlaagd worden.

Aanpassen:

```
2CDD 00 NOP
```

Gaat ook niet goed. Maar ook grappig. Gooit alleen nog maar wild barrels omdat hij steeds denkt dat hij met de eerste barrel bezig is. Kunnen twee dingen doen: de teller niet meer verlagen wanneer deze op #10 staat o.i.d. Of alle afhankelijkheden van de teller aanpassen. Eerst het eerste proberen.

```
2CDA 21B162 LD HL,#62B1 ; load bonus timer into HL
2CDD 7E LD A,(HL) ; load A with contents of HL
2CDE FE10 CP #10 ; bonus counter reached #10?
2CE0 2801 JR Z,#2CE2 ; yes, skip next step
2CE2 35 DEC (HL) ; decrease bonus timer
2CE3 000000 NOP, NOP, NOP
```

Ja dat gaat goed. Blijft oneindig doorlopen nu.

Nu versie v0.03 opgeslagen.

Nu ervoor zorgen dat jumpmans Y-positie niet gecorrigeerd wordt voor de schuine girders wanneer hij op de girder bij Pauline is.

Dit wordt gedaan door de volgende code:

```
2347 3EF0 LD A,#F0 ; A := #F0
2349 BD CP L ; is A == L ?
234A CA6023 JP Z,#2360 ; Yes, skip ahead

234D 3E4C LD A,#4C ; A := #4C
234F BD CP L ; == L ?
2350 CA6623 JP Z,#2366 ; yes, skip ahead

2353 7D LD A,L
2354 CB6F BIT 5,A
2356 CA5C23 JP Z,#235C
```

#4C is de jumpman Y-positie wanneer hij op de bovenste girder (Donkey Kong's girder) staat (het vlakke stuk). Dus in #234D zit een check daarop. Wellicht als volgt aanpassen:

```
234D C3C33F JP #3FC3 ; jump to additional code
2350 000000 NOP, NOP, NOP
```

en:

```
3FC3 3E30 LD A,#30 ; A := #30
3FC5 BD CP L ; == L ?
3FC6 C8 RET Z ; yes, on Pauline's girder: return
3FC7 3E4C LD A,#4C ; A := #4C
3FC9 BD CP L ; == L ?
3FCA CA6623 JP Z,#2366 ; yes, go back and skip part
3FCD C35323 JP #2353 ; go back
```

Nu versie v0.04 opgeslagen.

Even testen of de difficulty nog aangepast wordt. Difficulty wordt opgeslagen in #6380.

Ja die loopt gewoon op tot 5 en blijft dan op 5. Dat is niet goed. Kijken waar dat gezet wordt en daar dan NOP's van maken. Alleen met <P2> knop instelbaar maken.

De difficulty wordt maar op één plek gezet:

```

0398 FE05 CP #05 ; is this answer > 5 ?
039A 3802 JR C,#039E ; no, skip next step

039C 3E05 LD A,#05 ; otherwise A := 5

039E 328063 LD (#6380),A ; store result into difficulty
03A1 C9 RET ; return to #02DE

```

Aanpassen:

```
039E 000000 NOP, NOP, NOP
```

Nu staat de difficulty altijd op 0. Maar later in het begin dit goed zetten samen met de Wild frequentie. Voor nu deze wijziging eventjes nog niet doen: met difficulty 0 gebeuren er vreemde zaken.

Nu zodanig maken dat er vanuit het vrije slot van de main twee additionele procedures aangeroepen kunnen worden: beide knoppen apart afvangen.

```
19C2 CDD03F CALL #3FD0 ; call additional code
```

en:

```

3FD0 CDBBBB CALL BBBB ; call routine for handling <P1> button
3FD3 CDCCCC CALL CCCC ; call routine for handling <P2> button
3FD6 C9 RET ; return

```

Button <P1> om hamers en fireballs te resetten

#24ED t/m 25F1 kan verwijderd worden omdat dit alleen voor pies gebruikt wordt.

Daar kan dan additionele code komen

#6240 wordt gebruikt als timer

Aanpassen:

```
3FD0 CDED24 CALL 24ED ; call routine for handling <P1> button
```

en:

```

24ED 3A4062 LD A,(#6240) ; load A with timer
24F0 FE00 CP #00 ; is timer run out?
24F2 2805 JR Z,AAAA ; yes, check for P2 pressed – jump to label AAAA
;
24F4 3D DEC A ; decrement A
24F5 324062 LD (#6240),A ; write timer to memory
24F8 C9 RET ; return without check for P2 pressed

```

```

;
24F9 3A007D LD A,(IN2) ; load A with IN2 – label AAAA
24FC FE08 CP #04 ; is P2 button pressed?
24FE C0 RET NZ ; no, return without changing percentage
;
; Fireballs deactiveren
;
24FF 3E00 LD A,#00 ; load A with #00
2501 32D069 LD (#69D0),A ; move fireball 1 off screen
2504 32D469 LD (#69D4),A ; move fireball 1 off screen
2507 32D869 LD (#69D8),A ; move fireball 1 off screen
250A 32DC69 LD (#69DC),A ; move fireball 1 off screen
250D 32E069 LD (#69E0),A ; move fireball 1 off screen
2510 320064 LD (#6400),A ; deactivate fireball 1
2513 322064 LD (#6420),A ; deactivate fireball 2
2516 324064 LD (#6440),A ; deactivate fireball 3
2519 326064 LD (#6460),A ; deactivate fireball 4
251C 328064 LD (#6480),A ; deactivate fireball 5
;
251F C9 RET ; return

```

Dit gaat goed. Klopt het dat er een maximum is van vier fireballs in een barrel screen? Of is dit wellicht afhankelijk van de difficulty?

Nu versie v0.05 opgeslagen.

Nu de hamers restoren.

Lijkt erop alsof dit stukje code voldoende is. Hamers kennen geen datatabel. Grab wordt bepaald door collision met hammer sprite. Dus voldoende om de sprites weer te tekenen.

```

1000 210C3E LD HL,#3E0C ; load HL with table data for hammers on girders
1003 CDA611 CALL #11A6 ; ???

```

Aanpassen:

```

251F 210C3E LD HL,#3E0C ; load HL with table data for hammers on girders
2522 CDA611 CALL #11A6 ; ???
2525 C9 RET ; return

```

Ja dat gaat inderdaad goed.

Nu versie v0.06 opgeslagen.

Weergeven op het scherm van de Difficulty en de Wild frequentie.

Difficulty als D=01 i.p.v. L=01.

```

06D7 210375 LD HL,#7503 ; load HL with screen location for "L="
06DA 361C LD (HL),#1C ; draw "L"
06DC 21E374 LD HL,#74E3 ; next location
06DF 3634 LD (HL),#34 ; draw "="
06E1 3A2962 LD A,(#6229) ; load A with level #
06E4 fe64 CP #64 ; level < #64 (100 decimal) ?
06E6 3805 JR c,#06Ed ; yes, skip next 2 steps

```

Aanpassen:

```

06DA 3614 LD (HL),#14 ; draw "D"

```

en:

```

06E1 3A8063 LD A,(#6380) ; load difficulty in A
06E4 21C374 LD HL,#74C3 ; next location
06E7 77 LD (HL),A ; draw difficulty
06E8 C9 RET ; return

```

#06E6 t/m 06FD zijn nog vrij voor verdere uitwerking tekenen. En dat kan na update van Difficulty of Wild frequentie ook weer aangeroepen worden.

Gaat goed. **Merk met testen wel dat bij difficulty 5 en alle fireballs out wordt bij een reset wel de fireballs verwijderd, maar er komen geen blue barrels meer.**

Nu de W= gaan toevoegen.

Gebruiken #6245 voor vastleggen van de mode. Daar kan je dan doorheen lopen.

000X XXYY – XXX = difficulty en YY = wild frequentie.

XX 00 = All wilds (16x)
 01 = x2 = two times more wilds
 10 = x4 = four times more wilds
 11 = x8 = eight times more wilds

Dus doen:	AND met 0000 0011 (#03)	00	01	10	11
	INC	01	10	11	100
	Shift left 1x	10	100	110	1000
		02	04	06	08

Dat gaat niet goed.

Dan maar een routine maken die de mapping doet met CP's:

```

3FD7 3A4562 LD A,(#6245) ; load mode in A
3FDA FE01 CP #01 ; A == 1 ?
3FDC 2003 JR NZ,AAAA ; no, skip next steps

```

3FDE	3E02	LD	A,#02	; load A with 02
3FE0	C9	RET		; return
;				
3FE1	FE02	CP	#02	; A == 2 ?
3FE3	2003	JR	NZ,AAAA	; no, skip next steps
3FE5	3E04	LD	A,#04	; load A with 04
3FE7	C9	RET		; return XXXX
;				
3FE8	FE03	CP	#03	; A == 3 ?
3FEA	2003	JR	NZ,AAAA	; no, skip next steps
3FEC	3E08	LD	A,#08	; load A with 08
3FEE	C9	RET		; return XXXX
;				
3FEF	3EBF	LD	A,#BF	; load A with infinity character
3FF1	C9	RET		; return

En dan tekenen van de Wild frequency.

06E8	210275	LD	HL,#7502	; next location
06EB	3627	LD	(HL),#14	; draw "W"
06ED	210275	LD	HL,#74E2	; next location
06F0	3634	LD	(HL),#34	; draw "="
;				
06F2	CDD73F	CALL	#3FD7	; load A with wild factor
06F5	21C374	LD	HL,#74C2	; next location
06F8	77	LD	(HL),A	; draw difficulty
;				
06F9	C3F23F	JP	#3FF2	; jump to additional code

en:

3FF2	210275	LD	HL,#74A2	; next location
3FF5	3628	LD	(HL),#28	; draw "x"
3FF7	C9	RET		; return

Wordt nu goed weergegeven.

Nu versie v0.07 opgeslagen.

De W= nog van kleur veranderen. Ook blauw maken. Gedaan.

De afhandeling van <P2> gaan toevoegen.

#6241 wordt gebruikt als timer

Aanpassen:

3FD3	CD2625	CALL	2526	; call routine for handling <P2> button
------	--------	------	------	-----------------------------------------

en:

```
2526 3A4162 LD A,(#6241) ; load A with timer
2529 FE00 CP #00 ; is timer run out?
252B 2805 JR Z,#AAAA ; yes, check for P2 pressed – jump to label AAAA
;
252D 3D DEC A ; decrement A
252E 324162 LD (#6241),A ; write timer to memory
2531 C9 RET ; return without check for P2 pressed
;
2532 3A007D LD A,(IN2) ; load A with IN2 – label AAAA
2535 FE08 CP #08 ; is P2 button pressed?
2537 C0 RET NZ ; no, return without changing percentage
;
2538 3E10 LD A,#10 ; load A with decimal 16 (frames)
253A 324162 LD (#6241),A ; set timer
;
253D 3A4562 LD A,(#6245) ; load A with mode
2540 3C INC A ; increment A
2541 FE14 CP #14 ; a == 14 ( 0001 0100) ?
2543 2002 JR NZ,AAAA ; no, skip next step
2545 3E00 LD A,#00 ; reset mode
2547 324562 LD (#6245),A ; set mode
;
254A CDD706 CALL #06D7 ; update info on screen
;
254D C9 RET ; return
```

Gaat niet helemaal goed. De vaststelling van de W factor doet dit op basis van de hele waarde.
Vergeten om een AND #03 toe te voegen.

```
3FD7 3A4562 LD A,(#6245) ; load mode in A
3FDA FE01 CP #01 ; A == 1 ?
3FDC 2003 JR NZ,AAAA ; no, skip next steps
```

Tussenvoegen door aanpassen:

```
3FD7 C3F83F JP #3FF8 ; jump to additional code
```

en:

```
3FF8 3A4562 LD A,(#6245) ; load mode in A
3FFB E603 AND #03 ; make A value between 00 and 03
3FFD C3DA3F JP #3FDA ; jump back
```

Gaat nu goed.

Dan nu de Difficulty aanpassen.

254A	3A4562	LD	A,(#6245)	; load mode in A
254D	1F	RRA		; rotate right
254E	1F	RRA		; rotate right
254F	3C	INC	A	; increment
2550	E607	AND	#07	; mask to get difficulty
2552	328063	LD	(#6380),A	; set difficulty
;				
2555	CDD706	CALL	#06D7	; update info on screen
;				
2548	C9	RET		; return

Gaat ook goed.

Nu versie v0.08 opgeslagen.

Nu de wild factor zetten.

#6245	= mode	00(0)	01(1)	10(2)	11(3)
#6232	= wild factor	00	07	03	01

Routine maken om de wildfactor te bepalen

3F30	3A4562	LD	A,(#6245)	; load mode in A
3F33	E603	AND	#03	; mask to get factor
;				
3F35	FE01	CP	#01	; A == 1 ?
3F37	2003	JR	NZ,AAAA	; no, skip next steps
3F39	3E07	LD	A,#07	; load A with 02
3F3B	C9	RET		; return
;				
3F3C	FE02	CP	#02	; A == 2 ?
3F3E	2003	JR	NZ,AAAA	; no, skip next steps
3F30	3E03	LD	A,#03	; load A with 03
3F42	C9	RET		; return XXXX
;				
3F43	FE03	CP	#03	; A == 3 ?
3F45	2003	JR	NZ,AAAA	; no, skip next steps
3F47	3E01	LD	A,#01	; load A with 01
3F49	C9	RET		; return XXXX
;				
3F4A	3E00	LD	A,#00	; load A with 00
3F4C	C9	RET		; return

En aanroepen en factor zetten:

2555	CD303F	CALL	#3F30	; determine wild factor
------	--------	------	-------	-------------------------


```

;
XXXX 323262      LD      (#6232),A      ; set wild factor
;
XXXX CDD706      CALL    #06D7          ; update info on screen
;
XXXX C9          RET                    ; return

```

Dat gaat goed. Nu nog de wild routine aanpassen dat deze van de aangepaste wild factor gebruik gaat maken.

Aanpassen:

```

2C44 C34D3F      JP      #3F4D          ; jump to additional code
2C47 0000        NOP, NOP

```

En toevoegen code:

```

3F4D 213262      LD      HL,#6232      ; load HL with percentage address
3F50 46          LD      B,(HL)        ; load B with percentage
3F51 A0          AND     B             ; mask out its based on percentage
3F52 C2862C      JP      NZ,#2C86      ; If result is not zero, deploy a normal barrel.
3F55 C3492C      JP      #2C49        ; jump back

```

Dat gaat goed.

Ook nog eventjes het probleem met de bomb barrels oplossen.

Aanpassen:

```

2143 C3583F      JP      #3F58          ; jump to additional code

```

en:

```

3F58 3A1860      LD      A,(RngTimer1) ; load A with random number
3F5B E603        AND     #03          ; mask so A is value between 00 and 04
3F5D FE00        CP      #00          ; A = 00?
3F5F 2805        JR      Z,AAAA        ; yes, skip next steps – jump to label AAAA
;
3F61 3E00        LD      A,#00        ; load A with #00
3F63: DD7719      LD      (IX+#19),A   ; store A in IX+#19
;
3F66: C35321      JP      #2153        ; jump back – label AAAA

```

Gaat goed.

Nu versie v.0.09 opgeslagen.

Er zijn nog twee problemen geconstateerd.

- In bepaalde gevallen komen er geen fireballs meer na resetten.
- Het lijkt erop alsof de difficulty ook nog na een bepaalde tijd opgehoogd wordt.

Klopt de difficulty wordt op gegeven moment teruggezet naar 2. Op basis van het level lijkt dat. Dat is op #039E. Daar een NOP van maken.

```
039E  000000          NOP, NOP, NOP
```

Dat gaat niet goed. Dan wordt de difficulty in het begin op nul gezet. Dat oplossen door de 6380 bij opstarten op 01 te zetten.

Aanpassen:

```
0207  C3693F          JP      #3F69          ; jump to additional code
```

en:

```
3F69  3E01          LD      A,#01          ; load A with #01
3F6B  328063        LD      (#6380),A      ; set initial difficulty
3F6E  3A807D        LD      A,(DSW1)       ; load A with Dip switch settings
3F71  C30A02        JP      #020A         ; jump back
```

Ja dat werkt nu goed. Nu wordt er met difficulty 1 gestart.

Nee, want na starten wordt het weer teruggezet naar 0. Dat komt door een reset naar nul van een heleboel variabelen. Dit gebeurt in routine op #0F56 die aangeroepen wordt vanuit #0D5F. Dus de initiële configuratie moet ergens anders tussengevoegd worden.

```
0207  3A807D  LD      A,(DSW1)          ; load A with Dip switch settings
```

en:

```
0D5F  C3693F          JP      #3F69          ; jump to additional code
```

en:

```
3F69  CD560F        CALL   #0F56          ; clear and initialize RAM values
3F6C  3E01          LD      A,#01          ; load A with #01
3F6E  328063        LD      (#6380),A      ; set initial difficulty
3F71  C3620D        JP      #0D62         ; jump back
```

Gaat nu wel goed.

Weer gezien dat er geen blue barrels meer komen na reset van de fireballs. Dat nog eens bekijken qua code. Is er nog een afhankelijkheid bij het maken van een blue barrel? Mogelijk met de tijd

binnen het board (bonus time, die nu op een bepaalde waarde blijft 'hangen'. Maar wellicht bij die waarde geen blue barrels meer geven?).

Dat klopt. Dat is een samenspel tussen #62B1 en #62B2. Ze beginnen beiden op #32. Dan wordt #62B2 op #2A gezet en loopt #62B1 af (bij elke barrel deployment?). En wanneer ze weer gelijk zijn. Dan wordt er een blue barrel gedeployed. En dan wordt de #62B2 op #22 gezet en wordt er weer een blue barrel gedeployed als 62B1 die waarde bereikt. Maar dat houdt op bij #62B2 van #0A en #62B1 loopt niet lager dan #10 (is dat eerder voor deze hack begrenst?). Maar dat kan dus maar een keer en wordt niet herstart met een reset van de fireballs. Wellicht daarbij dan deze twee waarden ook resetten naar hun beginwaarden? En missen we nu geen vijfde fireball (daar lijkt het wel op).

Dus wellicht op te lossen door #62B1 op #32 en de #62B2 op #2A te zetten.

Dat moet dus na het resetten van de fireballs en de hamers gebeuren. Het laatste stuk daarvan is:

```
251F  210C3E      LD    HL,#3E0C      ; load HL with table data for hammers on girders
2522  CDA611      CALL  #11A6      ; ???
2525  C9          RET              ; return
```

Aanpassen:

```
2522  C3743F      JP    #3F74      ; jump to additional code
2525  00          NOP
```

en:

```
3F74  CDA611      CALL  #11A6      ; ???
;
3F77  3E32        LD    A,#32      ; load A with #32
3F79  32B162      LD    (#62B1),A  ; reset #62B1 (bonus timer)
3F7C  3E32        LD    A,#32      ; load A with #32
3F7E  32B262      LD    (#62B2),A  ; reset #62B2 (blue barrel timer)
;
3F81  C9          RET              ; return
```

Ja dat werkt goed.

Nu versie v0.10 opgeslagen.

Player 1 only maken

De teksten voor 1 player en 2 player worden hier getoond:

```

08D5 0604 LD B,#04 ; B := 4 = 0100 binary
08D7 1E09 LD E,#09 ; E := 9 , code for "ONLY 1 PLAYER BUTTON"
08D9 3A0160 LD A,(NumCredits) ; load A with number of credits
08DC FE01 CP #01 ; == 1 ?
08DE CAE408 JP Z,#08E4 ; yes, skip next 2 steps

08E1 060C LD B,#0C ; B := #0C = 1100 binary
08E3 1C INC E ; E := #0A, code for "1 OR 2 PLAYERS BUTTON"

08E4 3A1A60 LD A,(FrameCounter) ; load A with # Timer constantly counts down from FF to 00
08E7 E607 AND #07 ; mask bits. zero ?
08E9 C2F308 JP NZ,#08F3 ; no, skip next 3 steps

08EC 7B LD A,E ; yes, load A with E for code of text to draw, for buttons to press to start
08ED CDE905 CALL #05E9 ; draw text to screen
08F0 CD1606 CALL #0616 ; draw credits on screen

```

De tweede tekst weghalen door op #08DE een JP te doen i.p.v. JP Z.

Ja dat werkt.

En dan nu nog ervoor zorgen dat er geen 2 player game opgestart kan worden.

Dat wordt hier afgehandeld:

```

; jump from #08B5 when GameMode2 == 1

08F8 CDD508 CALL #08D5 ; draws press player buttons and loads A with IN2, masked by possible player numbers
08FB FE04 CP #04 ; is the player 1 button pressed ?
08FD CA0609 JP Z,#0906 ; yes, skip ahead

0900 FE08 CP #08 ; is the player 2 button pressed ?
0902 CA1909 JP Z,#0919 ; yes, skip ahead

0905 C9 RET ; return to #00D2

```

Dus geen two -player meer starten wanneer #0900 t/m #0904 in NOPS veranderd worden.

Dat klopt. Nu geen two-player mode meer.

En dan algemene teksten toevoegen door de default tekst van de High Score Table aan te passen.

WILDFINITY
USE THIS ROMHACK TO
PRACTICE WILD BARRELS

PRESS P1 TO RESET
PRESS P2 TO SELECT MODE

De default text voor de high score table staat hier:

```

3565: 94 77 01 23 24 10 10 00 00 07 06 05 00 10 10 10 10 10 10 10 10 10 10 10 3F 00 50 76 00 F4 76
3587: 96 77 02 1E 14 10 10 00 00 06 01 00 00 10 10 10 10 10 10 10 10 10 10 10 3F 00 00 61 00 F6 76
35A9: 98 77 03 22 14 10 10 00 00 05 09 05 00 10 10 10 10 10 10 10 10 10 10 10 3F 00 50 59 00 F8 76
35CB: 9A 77 04 24 18 10 10 00 00 05 00 05 00 10 10 10 10 10 10 10 10 10 10 10 3F 00 50 50 00 FA 76
35EE: 9C 77 05 24 18 10 10 00 00 04 03 00 00 10 10 10 10 10 10 10 10 10 10 10 3F 00 00 43 00 FC 76

```

Elke regel heeft 24 posities.

Op 3565: WILDFINITY (10 posities, 7 ervoor, 7 erachter)

27 19 1C 14 16 19 1E 19 24 29

Op 3587: USE THIS ROMHACK TO (18 posities , 3 ervoor, 3 erachter)

25 23 15 10 24 18 19 23 10 22 1F 1D 18 11 13 1B 10 24 1F

Op 35A9: PRACTICE WILD BARRELS (21 posities, 2 ervoor, 1 erachter)

20 22 11 13 24 19 13 15 10 27 19 1C 14 10 12 11 22 22 15 1C 23

Op 35CB: PRESS P1 TO RESET (17 posities, 4 ervoor, 3 erachter)

20 22 15 23 23 10 20 01 10 24 1F 10 22 15 23 15 24

Op 35EE: PRESS P2 TO SELECT MODE (23 posities, 1 ervoor, 0 erachter)

20 22 15 23 23 10 20 02 10 24 1F 10 23 15 1C 15 13 24 10 1D 1F 14 15

De header van de high score table ook leegmaken:

#37A0 t/m #37B0 alles spaties maken (10).

Nu versie v0.11 opgeslagen.

Na reviewopmerkingen van CrazyFox de volgende aanpassingen doorvoeren:

- Barreltraining scope: naam aanpassen naar iets pakkends.
- De 2x wilds vervangen door een 1x wilds en de volgorde veranderen: 1x, 4x, 8x, oneindig.
- Als P1 indrukken terwijl hamer actief, dan de actieve hamer afbreken.
- Visuele indicatie wanneer je geraakt wordt, bijvoorbeeld tijdelijke palette wijziging.

2x wilds vervangen door 1x wilds en de volgorde veranderen:

1^e mapping is voor de weergaven op het scherm:

3FD7	3A4562	LD	A,(#6245)	; load mode in A
3FDA	FE01	CP	#01	; A == 1 ?
3FDC	2003	JR	NZ,AAAA	; no, skip next steps

3FDE	3E04	LD	A,#04	; load A with 04
3FE0	C9	RET		; return
;				
3FE1	FE02	CP	#02	; A == 2 ?
3FE3	2003	JR	NZ,AAAA	; no, skip next steps
3FE5	3E08	LD	A,#08	; load A with 08
3FE7	C9	RET		; return XXXX
;				
3FE8	FE03	CP	#03	; A == 3 ?
3FEA	2003	JR	NZ,AAAA	; no, skip next steps
3FEC	3EBF	LD	A,#BF	; load A with infinity character
3FEE	C9	RET		; return XXXX
;				
3FEF	3E01	LD	A,#01	; load A with 01
3FF1	C9	RET		; return

2^e mapping is voor de wild barrel mask waarden:

3F30	3A4562	LD	A,(#6245)	; load mode in A
3F33	E603	AND	#03	; mask to get factor
;				
3F35	FE01	CP	#01	; A == 1 ?
3F37	2003	JR	NZ,AAAA	; no, skip next steps
3F39	3E03	LD	A,#03	; load A with 03
3F3B	C9	RET		; return
;				
3F3C	FE02	CP	#02	; A == 2 ?
3F3E	2003	JR	NZ,AAAA	; no, skip next steps
3F40	3E01	LD	A,#01	; load A with 01
3F42	C9	RET		; return XXXX
;				
3F43	FE03	CP	#03	; A == 3 ?
3F45	2003	JR	NZ,AAAA	; no, skip next steps
3F47	3E00	LD	A,#00	; load A with 00
3F49	C9	RET		; return XXXX
;				
3F4A	3E0F	LD	A,#0F	; load A with 0F
3F4C	C9	RET		; return

Gaat nog niet goed. De initieële waarde van de wild rate staat op 0 (en dat is dus oneindige barrels. Dus die moet op #0F geïnitieëld worden.

Huidige code om de difficulty te initialiseren uitbreiden met code om de wild rate ook te initialiseren.

3F69	CD560F	CALL	#0F56	; clear and initialize RAM values
3F6C	3E01	LD	A,#01	; load A with #01
3F6E	328063	LD	(#6380),A	; set initial difficulty
3F71	C3620D	JP	#0D62	; jump back

Aanpassen:

```
3F71  C3823F      JP      #3F82      ; jump to additional code
```

en:

```
3F82  3E0F      LD      A,#0F      ; load A with #0F
3F84  323262     LD      (#6232),A    ; set initial wild percentage
3F87  C3620D     JP      #0D62      ; jump back
```

Ja, gaat nu wel goed.

Als P1 ingedrukt dan de eventuele actieve hamer afbreken.

Als het goed is wordt #6217 gebruikt om aan te geven dat er een hamer actief is. Is het voldoende om deze terug op 0 te zetten? En wat gebeurt er dan met het eventuele hamer muziekje dat afgespeeld wordt. Kun je dat onderbreken?

Eerst maar eens proberen door de #6217 op #00 te zetten.

Het huidige laatste stuk van de afhandeling van P1 indrukken is:

```
3F77  3E32      LD      A,#32      ; load A with #32
3F79  32B162     LD      (#62B1),A    ; reset #62B1 (bonus timer)
3F7C  3E32      LD      A,#32      ; load A with #32
3F7E  32B262     LD      (#62B2),A    ; reset #62B2 (blue barrel timer)
;
3F81  C9        RET              ; return
```

Aanpassen:

```
3F7E  C38A3F      JP      #3F8A      ; jump to additional code
```

en:

```
3F8A  32B262     LD      (#62B2),A    ; reset #62B2 (blue barrel timer)
3F8D  3E32      LD      A,#00      ; load A with #00
3F8F  321762     LD      (#6217),A    ; reset #6217 (hammer activity)
;
3F92  C9        RET              ; return
```

Gaat in principe goed. Maar het hamermuziekje blijft oneindig lang doorspelen. Dus dat wordt normaal aan het einde van de hamer ook expliciet uitgezet. Kijken hoe dat gedaan wordt en dat toevoegen.

Er blijkt ook een heel stuk code te zijn dat uitgevoerd wordt wanneer een hamer afloopt. Kan dat dan niet gewoon aangeroepen worden?

Aanpassen:

```
3F8A  32B262      LD      (#62B2),A      ; reset #62B2 (blue barrel timer)
3F8D  CD5A2F      CALL     #2F5A        ; call routine to end hammer
3F90  C9          RET              ; return
```

Ja dat gaat goed.

Dan kijken hoe in de romhacks van sockmaster visueel weergegeven wordt dat een hamer gebruikt wordt. Ja in Spooky Remix wordt dat gedaan. En dat lijkt inderdaad op een change van palettes.

Eerst kijken waar we het doodgaan afgevangen hebben.

In plaats van het aantal levens terug te zetten (in regel #2819) wordt nu gesprongen naar een stukje code dat het doodgaan geluid afspeelt en dan returned. Nu dat iet doen. Nu vanuit #2819 springen naar sen stuk additionele code waarin het palette aangepast wordt en een timer (#6233) op een bepaalde waarde gezet wordt.

Aanpassen:

```
2819  C35F25      JP      #255F        ; jump to additional code
```

en:

```
255F  3E20        LD      A,#20
2561  323362      LD      (#6233),A      ; set palette timer
;
2564  C3A612      JP      #12A6        ; play death sound, and return
```

En de afhandeling van deze timer als extra routine aan de main toevoegen.

Additioneel slot toevoegen:

```
19C2  CD6725      CALL     #2567        ; call additional code
```

en (waarbij op #3FD0 de sloten voor indrukken P1 en P2 afgehandeld worden):

```
2567  CD6E25      CALL     256E        ; call routine for handling palette changes
256A  CDD03F      CALL     3FD0        ; call routines for handling <P1> and <P2>
256D  C9          RET              ; return
```

en:

```
256E  3A3362      LD      A,(#6233)    ; load A with palette timer
2571  FE00        CP      #00        ; A == 00 ?
2573  C8          RET      Z          ; yes, return, no timer set
;
```


2574	3D	DEC	A	; decrement A
2575	323362	LD	(#6233),A	; store in palette timer
;				
2578	FE20	CP	#19	; A == 19 ? (start timer)
257A	2805	JR	Z,#2581	; yes, jump to label AAAA
;				
257C	FE00	CP	#00	; A == 00 ? (end timer)
257E	280C	JR	Z,#258C	; yes, jump to label BBBB
;				
2580	C9	RET		; return
;				
2581	3E01	LD	A,#01	; label AAAA
2583	32867D	LD	(REG_PALETTE_A),A	; set palette A
2586	3E00	LD	A,#00	
2588	32877D	LD	(REG_PALETTE_B),A	; set palette B
258B	C9	RET		; return
;				
258C	3E00	LD	A,#00	; label BBBB
258E	32867D	LD	(REG_PALETTE_A),A	; set palette A
2591	3E01	LD	A,#01	
2593	32877D	LD	(REG_PALETTE_B),A	; set palette B
2596	C9	RET		; return

Ja dat werkt.

Nu alleen nog een goede titel bedenken en dan aanpassing bovenaan scherm, op titel scherm en op het high score scherm.

BARRELBOS
 BARRELMASER
 BARRELEARN
 BARRELLEARNER
 BARRELTRAIN
 BARRELB∞SS

BARRELB∞SS wordt het: 12 11 22 22 15 1C 12 BF 23 23

Aanpassen titel bovenin:

#36B2: 60 76 12 11 22 22 15 1C 12 BF 23 23 3F

Aanpassen titel in title screen:

En dan in #3F18 de titel maken:
 WILDFINITY

#3F18: 8E 76 12 11 22 22 15 1C 12 BF 23 23 3F

En dan in het info scherm (voorheen high score screen):

Op 3565: BARRELB∞SS vervangen door (10 posities, 7 ervoor, 7 erachter)

12 11 22 22 15 1C 12 BF 23 23

Nu versie v0.12 opgeslagen.

Dit is Final versie v1.00.

Nog een probleem dat de hamer niet gedeactiveerd wordt bij reset wanneer actief en daardoor punten blijft geven voor elke wild barrel die de teruggehangen hamer raakt.

Hamer deactiveren werd eerst gedaan door:

```
3F8D  3E32      LD    A,#00      ; load A with #00
3F8F  321762     LD    (#6217),A  ; reset #6217 (hammer activity)
```

Maar later vervangen door aanroep naar code die de hamer deactiveert:

```
3F8D  CD5A2F     CALL  #2F5A      ; call routine to end hammer
3F90  C9         RET             ; return
```

Maar wordt in de code vanaf #2F5A wel de #6217 (hammer activity) ook op #00 gezet?

Ja dat wordt wel gedaan. Maar wel een beetje de vraag of het wel goed gaat. Er wordt namelijk ook gelezen van en geschreven naar adressen gerelateerd aan de IX, maar die zet ik niet. Dus dat is wel gevaarlijk.

Bij normaal aflopen onderste hamer is de IX #6690 en bij de bovenste hamer is de IX #6680.

Misschien op te lossen door twee calls naar #2F5A te doen: één keer met #6680 in IX en één keer met #6690 in IX. In dat stukje routine de IX push'en en pop'en. Zodat de tijdelijke aanpassing van de reset niet iets anders in de war gooit.

Aanpassen:

```
3F8D  C39725     JP     #2597      ; jump to additional code
3F90  00         NOP
```

En:

```
2597  DDE5       PUSH  IX          ; store IX for later
2599  DD218066    LD    IX,#6680    ; set IX top hammer
259D  CD5A2F     CALL  #2F5A      ; call routine to end hammer
25A0  DD219066    LD    IX,#6690    ; set IX bottom hammer
25A4  CD5A2F     CALL  #2F5A      ; call routine to end hammer
25A7  DDE1       POP   IX          ; restore IX
25A9  C9         RET             ; return
```

Dat gaat niet goed. De hamer geeft inderdaad geen hits meer wanneer deze is teruggehangen. Maar de hamer kan ook niet meer gepakt worden.

Op basis van debugging lijkt het alsof de 6680 en 6690 op #00 blijven staan (de IX+#00), terwijl deze bij een nieuw spel initieel op #01 staan. Maar dat wordt juist gezet door de routine op #2F5A. Daarmee wordt waarschijnlijk aangegeven dat de hamer gebruikt is.

Dus deze eventjes ook nog terug op #01 zetten.

Aanpassen:

```

2597 DDE5      PUSH IX          ; store IX for later
;
2599 DD218066  LD    IX,#6680    ; set IX top hammer
259D CD5A2F    CALL  #2F5A        ; call routine to end hammer
25A0 DD360001  LD    (IX+#00),#01 ; make top hammer available again
;
25A4 DD219066  LD    IX,#6690    ; set IX bottom hammer
25A8 CD5A2F    CALL  #2F5A        ; call routine to end hammer
25AB DD360001  LD    (IX+#00),#01 ; make bottom hammer available again
;
25AF DDE1      POP  IX          ; restore IX
25B1 C9        RET              ; return

```

Werkt nog steeds niet.

Wat ook vreemd is, is dat wanneer je zonder hamer pakken de P1 indrukt dat nu de achtergrondmuziek helemaal verdwijnt. Dat kan ook niet de bedoeling zijn.

Misschien even kijken hoe de hamers aan het begin geïnitieerd worden en dat opnieuw aanroepen of nabouwen?

Hamers plaatsen lijkt te gebeuren door routine vanaf #11A6. En dat wordt voor hamers aangeroepen vanuit #1000:

```

1000 210C3E LD    HL,#3E0C      ; load HL with table data for hammers on girders
1003 CDA611 CALL  #11A6        ; ???

```

Dus wellicht opnieuw initialiseren door dit aan te roepen?

Aanpassen:

```

2597 3E00      LD    A,#00      ; load A with #00
2599 321762    LD    (#6217),A   ; reset #6217 (hammer activity)
259C 328166    LD    (#6681),A   ; set top hammer inactive
259F 329166    LD    (#6691),A   ; set bottom hammer inactive
25A2 210C3E    LD    HL,#3E0C    ; load HL with table data for hammers on girders
25A5 CDA611    CALL  #11A6        ; recreate hammers
25A8 C9        RET              ; return

```

Lijkt te werken, alleen muziekje moet weg. Maar die herstelt de oude muziek. Maar als je dat doet als er geen hamer actief is, dan verdwijnt de normale muziek. Dus eerste checken of er een hamer actief is.

```

2597  3A1762      LD      A,(#6217)      ; load A with hammer active indicator
259A  FE01        CP      #01           ; is hammer active?
259C  2011        JR      NZ,#AAAA      ; no, skip next steps – jump to label AAAA
;
259E  3E00        LD      A,#00         ; load A with #00
25A0  321762      LD      (#6217),A     ; reset #6217 (hammer activity)
25A3  328166      LD      (#6681),A     ; set top hammer inactive
25A6  329166      LD      (#6691),A     ; set bottom hammer inactive
25A9  3A8963      LD      A,(#6389)     ; load A with previous background music
25AC  328960      LD      (#6089),A     ; set music with what it was before the hammer
;
25AF  210C3E      LD      HL,#3E0C      ; load HL with table data for hammers – label AAAA
25B2  CDA611      CALL   #11A6         ; recreate hammers
25B5  C9          RET                  ; return

```

Dat lijkt nu goed te gaan.

Zit nog een fout in de BARRELB∞SS titel in het info scherm. Er staat ABRRELB∞SS.

En dan nog versie aanpassen naar V1.01:

```
#373E:  3D 76 26 01 2B 00 01 10 3F (V1.01)
```

Nu versie v1.00.1 opgeslagen.

Gaat nog steeds niet goed. Wanneer P1 reset wanneer de hamer geel is, dan wordt als de hamer snel gepakt wordt, alleen het reserende deel van de oude, onderbroken, hamertijd gedaan en wordt het hameren veel te snel afgebroken. Dat is blijkbaar ook nog een setting die nog gereset moet worden.

De hammer length wordt opgeslagen in #6395. Even met debugger kijken wat er gebeurt met deze waarde wanneer er een P1 gedaan wordt. #6395 is 0 als de hamer niet gebruikt wordt en als de hamer gebruikt wordt in de eerste fase. #6395 is 1 als de hamer gebruikt wordt in de tweede fase en van kleur veranderd is. Wordt dan opgehoogd naar 2 en dat is de trigger om de hamer te eindigen. Als P1 gedrukt wordt, dan blijft de waarde op 1 staan en dat heeft impact op de volgende hamer.

Kijken of het opgelost kan worden door de #6395 ook te resetten.

```

2597  3A1762      LD      A,(#6217)      ; load A with hammer active indicator
259A  FE01        CP      #01           ; is hammer active?
259C  2011        JR      NZ,#AAAA      ; no, skip next steps – jump to label AAAA
;
259E  3E00        LD      A,#00         ; load A with #00

```

25A0	321762	LD	(#6217),A	; reset #6217 (hammer activity)
25A3	329563	LD	(#6395),A	; reset #6395 (hammer length)
25A6	328166	LD	(#6681),A	; set top hammer inactive
25A9	329166	LD	(#6691),A	; set bottom hammer inactive
25AC	3A8963	LD	A,(#6389)	; load A with previous background music
25AF	328960	LD	(#6089),A	; set music with what it was before the hammer
;				
25B2	210C3E	LD	HL,#3E0C	; load HL with table data for hammers – label AAAA
25B5	CDA611	CALL	#11A6	; recreate hammers
25B8	C9	RET		; return

Maar ook nog zelfde probleem met hamer terughangen in de eerste fase. Dan bij opnieuw pakken van de hamer is de tijd korter. Er is ook nog een hammer timer in #6394. Met debugger kijken wat daarmee gebeurt. Ja die moet dus ook nog gereset worden.

2597	3A1762	LD	A,(#6217)	; load A with hammer active indicator
259A	FE01	CP	#01	; is hammer active?
259C	2011	JR	NZ,#AAAA	; no, skip next steps – jump to label AAAA
;				
259E	3E00	LD	A,#00	; load A with #00
25A0	321762	LD	(#6217),A	; reset #6217 (hammer activity)
25A3	329463	LD	(#6394),A	; reset #6395 (hammer timer)
25A6	329563	LD	(#6395),A	; reset #6395 (hammer length)
25A9	328166	LD	(#6681),A	; set top hammer inactive
25AC	329166	LD	(#6691),A	; set bottom hammer inactive
25AF	3A8963	LD	A,(#6389)	; load A with previous background music
25B2	328960	LD	(#6089),A	; set music with what it was before the hammer
;				
25B5	210C3E	LD	HL,#3E0C	; load HL with table data for hammers – label AAAA
25B8	CDA611	CALL	#11A6	; recreate hammers
25BB	C9	RET		; return

Nu gaat het wel helemaal goed.

Nu versie v1.00.2 opgeslagen.

Deze versie is de finale versie v1.01.